
This is the second installment of an exposition of an ACL2 formalization of elementary linear algebra.
It extends the results of Part I, which covers the algebra of matrices over a commutative ring, but
focuses on aspects of the theory that apply only to matrices over a field: elementary row reduction
and its application to the computation of matrix inverses and the solution of simultaneous systems of
linear equations.

1 Introduction

This is the second installment of an exposition of an ACL2 formalization of elementary linear algebra.
Part I [6], which is also included in this workshop, covers the algebra of matrices over a commutative
ring with unity and their determinants. In this sequel, we focus on aspects of the theory of matrices
that apply only to matrices over a field, i.e., depend on the existence of a multiplicative inverse operator.
These include row reduction and its application to matrix invertibility and the solution of systems of
linear equations. In an anticipated Part III, all of these results will be applied to the study of abstract
vector spaces and linear transformations.

The proof scripts supporting both papers reside in the same directory, books/projects/linear/.
As described in [6], the abstract definition of a ring is formalized in the file ring.lisp by a set of
constrained encapsulated functions: a predicate rp that recognizes ring elements, the binary addition and
multiplication operations r+ and r*, the corresponding identity constants r0 and r1, and the additive
inverse operator r-. The notion of a field is similarly defined by an encapsulation in the file field.lisp,
in which these functions are renamed fp, f+, f*, etc.:

(encapsulate (((fp *) => *) ;field element recognizer

((f+ * *) => *) ((f* * *) => *) ;addition and multiplication

((f0) => *) ((f1) => *) ;identities

((f- *) => *) ((f/ *) => *)) ;inverses

(local (defun fp (x) (rationalp x)))

(local (defun f+ (x y) (+ x y)))

(local (defun f* (x y) (* x y)))

(local (defun f0 () 0))

(local (defun f1 () 1))

(local (defun f- (x) (- x)))

(local (defun f/ (x) (/ x)))

;; Closure:

...

;; Multiplicative inverse:

(defthm fpf/

(implies (and (fp x) (not (equal x (f0)))) (fp (f/ x))))

(defthm f*inv

(implies (and (fp x) (not (equal x (f0)))) (equal (f* x (f/ x)) (f1)))))

The only other difference between the two encapsulations is the inclusion here of the multiplicative
inverse f/ along with two constraining axioms, appended to the constraints adapted from the ring axioms.

Informally we shall refer to the field F that is characterized by this encapsulation. When our intention
is clear, the identity elements (f0) and (f1) will be abbreviated as 0 and 1. Clearly, all properties of
rings hold for fields as well. Thus, all definitions and theorems that appear in ring.lisp, rmat.lisp,
and rdet.lisp have analogs in the corresponding files field.lisp, fmat.lisp, and fdet.lisp. In
particular, the predicate flistnp recognizes a list of specified length of elements of F, called an flist;
fmatp recognizes a matrix over F; and fdet computes its determinant. In principle, all results in the latter
set of files could be derived by functional instantiation from the corresponding events in the former, but
we found it more expedient to reproduce the proofs, simply by selectively replacing occurrences of the
character r with f. An additional file, reduction.lisp, contains the results reported in this paper.
In describing these results, we assume the reader is familiar with Part I and is aware of the renaming
convention.

In Section 2, we define the notion of a reduced row-echelon matrix and develop a procedure that
converts an arbitrary matrix to reduced row-echelon form. An equivalent procedure, based on matrix
multiplication, is also defined. This leads to a criterion for invertibility of a square matrix and a method
for computing inverses. A second method of matrix inversion, based on determinants and the classical
adjoint, is derived from the results of Part I.

Section 3 addresses the solution of systems of linear equations, mainly as an application of row
reduction. We derive algorithmic tests for solvability and uniqueness of the solution, as well as a formula
that computes the solution in the uniquely solvable case. For the special case of an invertible square
coefficient matrix, we prove Cramer’s Rule, an alternative formula based on determinants. In the general
solvable case, we show that the solution set is infinite and establish a test that identifies solutions. In
Part III, this will lead to a formula that generates a basis for the solution space of a homogeneous system
of equations.

All of these results, which are stated and proved in the context of an abstract field, may be applied
to any concrete field of interest through functional instantiation. Eventually, we plan to apply the theory
to algebraic number fields. Of course, in their abstract formulation based on constrained functions,
the definitions are not executable. For the immediate purposes of illustration and testing, however, all
functions defined in field.lisp, fmat.lisp, fdet.lisp, and reduction.lisp have been adapted
to the field of rational numbers as executable functions, which are listed in the file rational.lisp.

2 Row Reduction

2.1 Reduced Row-Echelon Form

A reduced row-echelon matrix may be characterized as follows:

(1) Every all-zero row is preceded by every nonzero row;

(2) The first nonzero entry of each nonzero row is 1, and every other entry in the same column is 0;

(3) The column of the leading 1 in the ith nonzero row is an increasing function of i.

The formalization of this definition requires several auxiliary functions. First, we define the index of the
leading nonzero entry of a nonzero row r:

(defun first-nonzero (r)

(if (consp r)

(if (= (car r) (f0))

(1+ (first-nonzero (cdr r)))

0)

()))

In the following, we assume that a is an m×n matrix. Starting with row k, where 0 ≤ k ≤ m, find the row
of a with nonzero entry of least index, or return NIL if all rows beyond the first k are 0:

(defun row-with-nonzero-at-least-index (a m k)

(if (and (natp k) (natp m) (< k m))

(let ((i (row-with-nonzero-at-least-index a (1- m) k)))

(if (or (flist0p (nth (1- m) a))

(and i (<= (first-nonzero (nth i a)) (first-nonzero (nth (1- m) a)))))

i

(1- m)))

()))

Given j < n and k < m, check that (entry k j a) = 1 and that all other entries in column k are 0:

(defun column-clear-p (a k j m)

(if (zp m) t

(and (= (nth j (nth (1- m) a)) (if (= (1- m) k) (f1) (f0)))

(column-clear-p a k j (1- m)))))

Given k ≤ m, check that the first k rows of a form a reduced row-echelon matrix:

(defun row-echelon-p-aux (a m k)

(if (zp k) t

(and (row-echelon-p-aux a m (1- k))

(let ((i (row-with-nonzero-at-least-index a m (1- k))))

(or (null i)

(and (= i (1- k))

(column-clear-p a i (first-nonzero (nth i a)) m)))))))

Finally, check that a is a reduced row-echelon matrix:

(defund row-echelon-p (a) (row-echelon-p-aux a (len a) (len a)))

2.2 Conversion to Reduced Row-Echelon Form

We shall develop a procedure that converts an arbitrary m×n matrix a to reduced row-echelon form by a
sequence of elementary row operations of three types:

(1) Multiply row k by a scalar c:

(defund ero1 (a c k) (replace-row a k (flist-scalar-mul c (nth k a))))

(2) Add a scalar multiple of row j to row k, where j ̸= k:

(defund ero2 (a c j k)

(replace-row a k (flist-add (flist-scalar-mul c (nth j a)) (nth k a))))

(3) Interchange rows i and j, where i ̸= j:

(defund ero3 (a j k) (replace-row (replace-row a k (nth j a)) j (nth k a)))

Under the assumption that (entry k j a) = 1, the following function applies ero2 to clear all other
entries in column j by adding the appropriate multiple of row k to each of the other rows:

(defun clear-column (a k j m)

(if (zp m) a

(if (= (1- m) k)

(clear-column a k j (1- m))

(clear-column (ero2 a (f- (nth j (nth (1- m) a))) k (1- m))

k j (1- m)))))

Assume the first k rows of a are in reduced row-echelon form, i.e., (row-echelon-p-aux a m k)

= T, where k < m, and that i = (row-with-nonzero-at-least-index a m k) ̸= NIL. Let j =
(first-nonzero (nth i a)). The following function performs the next step of the reduction, pro-
ducing a matrix a’ satisfying (row-echelon-p-aux a’ m (1+ k)):

(defund row-reduce-step (a m k i j)

(clear-column (ero3 (ero1 a (f/ (nth j (nth i a))) i)

i k)

k j m))

The function row-reduce converts a to a reduced row-echelon matrix, using an auxiliary function that
completes the conversion under the assumption (row-echelon-p-aux a m k), where 0 ≤ k ≤ m:

(defun row-reduce-aux (a m k)

(let ((i (row-with-nonzero-at-least-index a m k)))

(if (and (natp k) (natp m) (< k m) i)

(row-reduce-aux (row-reduce-step a m k i (first-nonzero (nth i a)))

m (1+ k))

a)))

(defund row-reduce (a) (row-reduce-aux a (len a) 0))

The following confirms that this procedure produces the desired result:

(defthmd row-echelon-p-row-reduce

(implies (and (natp m) (natp n) (fmatp a m n))

(row-echelon-p (row-reduce a))))

We also note that row reduction does not alter a reduced row-echelon matrix:

(defthmd row-reduce-row-echelon-p

(implies (and (posp m) (posp n) (fmatp a m n) (row-echelon-p a))

(equal (row-reduce a) a)))

As an example, consider the following 4×5 matrix (a0):

DM !>(defun a0 () ’((0 -3 -6 4 9) (-1 -2 -1 3 1) (-2 -3 0 3 -1) (1 4 5 -9 -7)))

In the first step in the row reduction of (a0), row 1 is divided by its leading nonzero entry, -1, and
interchanged with row 0. The other entries in column 0 are then cleared:

DM !>(row-reduce-step (a0) 4 0 1 0)

((1 2 1 -3 -1)

(0 -3 -6 4 9)

(0 1 2 -3 -3)

(0 2 4 -6 -6))

Row reduction of (a0) requires three executions of row-reduce-step:
DM !>(row-reduce (a0))

((1 0 -3 0 5)

(0 1 2 0 -3)

(0 0 0 1 0)

(0 0 0 0 0))

We define the row rank of a to be the number of nonzero rows of (row-reduce a):
(defun num-nonzero-rows (a)

(if (consp a)

(if (flist0p (car a)) 0 (1+ (num-nonzero-rows (cdr a))))

0))

(defun row-rank (a) (num-nonzero-rows (row-reduce a)))

Note that (row-reduce (a0)) has 3 nonzero rows:
DM !>(row-rank (a0))

3

Obviously, the row rank of an m×n matrix cannot exceed m:
(defthmd row-rank<=m

(implies (and (fmatp a m n) (posp m) (posp n))

(<= (row-rank a) m)))

Nor can the row rank exceed n. To see this, consider the list of indices of the leading 1s of the nonzero
rows of a reduced row-echelon matrix a:
(defun lead-inds (a)

(if (and (consp a) (not (flist0p (car a))))

(cons (first-nonzero (car a)) (lead-inds (cdr a)))

()))

DM !>(lead-inds (row-reduce (a0)))

(0 1 3)

Clearly, the length of (lead-inds a) is the number of nonzero rows of a. Furthermore, (lead-inds
a) is a strictly increasing sublist of (ninit t). It follows that (len (lead-inds a)) ≤ n. Conse-
quently, the row rank of a is bounded by n:
(defthmd row-rank<=n

(implies (and (fmatp a m n) (posp m) (posp n))

(<= (row-rank a) n)))

We also note that if (row-rank a) = n, then (lead-inds a) is an increasing sublist of (ninit n) of
length n, which implies that the two lists are equal:
(defthmd lead-inds-ninit

(implies (and (fmatp a m n) (posp m) (posp n)

(row-echelon-p a) (= (row-rank a) n))

(equal (lead-inds a) (ninit n))))

Along with row reduction, there is an obvious analogous notion of column reduction and a corre-
sponding definiton of the column rank of a matrix, which may alternatively be defined as the row rank of
its transpose. As we shall show in Part III in the context of vector spaces, the row and column ranks of a
matrix are always equal.

2.3 Row Reduction as Matrix Multiplication

Once we have identified the sequence of operations required to derive the reduced row-echelon form of
an m × n matrix a, an alternative derivation may be achieved by applying the same operations to the m

× m identity matrix and right-multiplying the result by a. To this end, a row operation is encoded as a
list of length 3 or 4; the first member indicates the operation type (1, 2, or 3 as listed in the preceding
subsection), and the others are the parameters of the operation. The following predicate characterizes an
encoding of a row operation on a matrix of m rows:

(defund row-op-p (op m)

(and (true-listp op)

(case (car op)

(1 (and (= (len op) 3) (fp (cadr op)) (not (= (cadr op) (f0)))

(natp (caddr op)) (< (caddr op) m)))

(2 (and (= (len op) 4) (fp (cadr op))

(natp (caddr op)) (< (caddr op) m)

(natp (cadddr op)) (< (cadddr op) m)

(not (= (caddr op) (cadddr op)))))

(3 (and (= (len op) 3) (natp (cadr op)) (< (cadr op) m)

(natp (caddr op)) (< (caddr op) m))))))

The function apply-row-op applies an encoded row operation to a matrix:

(defund apply-row-op (op a)

(case (car op)

;(apply-row-op (list 1 c k) a) = (ero1 a c k)

(1 (ero1 a (cadr op) (caddr op)))

;(apply-row-op (list 2 c j k) a) = (ero2 a c j k)

(2 (ero2 a (cadr op) (caddr op) (cadddr op)))

;(apply-row-op (list 3 j k) a) = (ero3 a j k)

(3 (ero3 a (cadr op) (caddr op)))))

A list of row operations is identified in the obvious way:

(defun row-ops-p (ops m)

(if (consp ops)

(and (row-op-p (car ops) m)

(row-ops-p (cdr ops) m))

(null ops)))

The function apply-row-ops applies a list of operations in sequence from left to right:

(defun apply-row-ops (ops a)

(if (consp ops)

(apply-row-ops (cdr ops) (apply-row-op (car ops) a))

a))

By examining the definitions of row-reduce and its auxiliary functions, we construct the list of en-
codings of the operations that reduce a matrix a to reduced row-echelon form. The next four functions
encode the lists of operations performed by clear-column, row-reduce-step, row-reduce-aux, and
row-reduce, respectively:

(defun clear-column-ops (a k j m)

(if (zp m) ()

(if (= k (1- m))

(clear-column-ops a k j (1- m))

(cons (list 2 (f- (nth j (nth (1- m) a))) k (1- m))

(clear-column-ops (ero2 a (f- (nth j (nth (1- m) a))) k (1- m))

k j (1- m))))))

(defund row-reduce-step-ops (a m k i j)

(cons (list 1 (f/ (nth j (nth i a))) i)

(cons (list 3 i k)

(clear-column-ops (ero3 (ero1 a (f/ (nth j (nth i a))) i) i k)

k j m))))

(defun row-reduce-aux-ops (a m k)

(let* ((i (row-with-nonzero-at-least-index a m k))

(j (and i (first-nonzero (nth i a)))))

(if (and (natp k) (natp m) (< k m) i)

(append (row-reduce-step-ops a m k i j)

(row-reduce-aux-ops (row-reduce-step a m k i j) m (1+ k)))

(defund row-reduce-ops (a) (row-reduce-aux-ops a (len a) 0))

The correctness of this encoding procedure is confirmed by the following:

(defthmd apply-row-reduce-ops

(implies (and (fmatp a m n) (posp m) (posp n))

(equal (apply-row-ops (row-reduce-ops a) a)

(row-reduce a))))

Returning to the example of Subsection 2.2, we find that the first step in the row reduction of (a0)
involves five elementary operations:

DM !>(row-reduce-step-ops (a0) 4 0 1 0)

((1 -1 1) (3 1 0) (2 -1 0 3) (2 2 0 2) (2 0 0 1))

DM !>(apply-row-ops ’((1 -1 1) (3 1 0) (2 -1 0 3) (2 2 0 2) (2 0 0 1)) (a0))

((1 2 1 -3 -1)

(0 -3 -6 4 9)

(0 1 2 -3 -3)

(0 2 4 -6 -6))

The reader may wish to compute (row-reduce-ops (a0)), a list of length 15, and check that the
lemma apply-row-reduce-ops holds in this case.

The m×m elementary matrix corresponding to a row operation is defined to be the result of applying
the operation to the m×m identity matrix:

(defund elem-mat (op m) (apply-row-op op (id-fmat m)))

Application of a row operation is equivalent to left multiplication by the corresponding elementary ma-
trix:

(defthmd elem-mat-row-op

(implies (and (fmatp a m n) (row-op-p op m) (posp m) (posp n))

(equal (fmat* (elem-mat op m) a) (apply-row-op op a))))

The product of the list of elementary matrices associated with the row reduction of a matrix is computed
recursively by the function row-reduce-mat:

(defund row-ops-mat (ops m)

(if (consp ops)

(fmat* (row-ops-mat (cdr ops) m) (elem-mat (car ops) m))

(id-fmat m)))

(defund row-reduce-mat (a) (row-ops-mat (row-reduce-ops a) (len a)))

It follows from elem-mat-row-op by induction that applying a sequence ops of row operations to a is
equivalent to multiplication of a by (row-ops-mat ops m):

(defthmd fmat*-row-ops-mat

(implies (and (fmatp a m n) (posp m) (posp n)

(row-ops-p ops m))

(equal (fmat* (row-ops-mat ops m) a)

(apply-row-ops ops a))))

In particular, by apply-row-reduce-ops, row reduction of a is equivalent to multiplication by (row-

reduce-mat a):

(defthmd row-ops-mat-row-reduce

(implies (and (fmatp a m n) (posp m) (posp n))

(equal (fmat* (row-reduce-mat a) a) (row-reduce a))))

In our example, the product of the 15 elementary matrices corresponding to (row-reduce-ops (a0))

is

DM !>(row-reduce-mat (a0))

(( 3/5 -3/5 -1/5 0)

(-3/5 8/5 -4/5 0)

(-1/5 6/5 -3/5 0)

( 0 5 -2 1))

The conclusion of the lemma row-ops-mat-row-reduce may be readily verified for this case.

2.4 Invertibility

In this subsection, we focus on square matrices. Given an n×n matrix a, we seek an inverse of a, i.e., an
n×n matrix b such that

(fmat* a b) = (fmat* b a) = (id-fmat n).

If such a matrix exists, then it is unique in the strong sense that it is the only left or right inverse of a. For
example, if (fmat* c a) = (id-fmat n), then

c = (fmat* c (id-fmat n))

= (fmat* c (fmat* a b))

= (fmat* (fmat* c a) b))

= (fmat* (id-fmat n) b))

= b,

and the same conclusion similarly follows from the assumption (fmat* a c) = (id-fmat n). Thus,
we have

(defthm inverse-unique

(implies (and (fmatp a n n) (fmatp b n n) (fmatp c n n) (posp n)

(= (fmat* a b) (id-fmat n)) (= (fmat* b a) (id-fmat n))

(or (= (fmat* a c) (id-fmat n)) (= (fmat* c a) (id-fmat n))))

(equal c b)))

Every elementary matix has an inverse:

(defund invert-row-op (op)

(case (car op)

(1 (list 1 (f/ (cadr op)) (caddr op)))

(2 (list 2 (f- (cadr op)) (caddr op) (cadddr op)))

(3 op)))

(defthmd fmat*-elem-invert-row-op

(implies (and (row-op-p op n) (posp n))

(and (equal (fmat* (elem-mat (invert-row-op op) n) (elem-mat op n))

(id-fmat n))

(equal (fmat* (elem-mat op n) (elem-mat (invert-row-op op) n))

(id-fmat n)))))

Consequently, every product of elementary matrices has an inverse:

(defun invert-row-ops (ops)

(if (consp ops)

(append (invert-row-ops (cdr ops)) (list (invert-row-op (car ops))))

()))

(defthmd invert-row-ops-mat

(implies (and (row-ops-p ops n) (posp n))

(and (equal (fmat* (row-ops-mat (invert-row-ops ops) n)

(row-ops-mat ops n))

(id-fmat n))

(equal (fmat* (row-ops-mat ops n)

(row-ops-mat (invert-row-ops ops) n))

(id-fmat n)))))

We shall show that a has an inverse iff (row-rank a) = n and that in this case, the inverse of a is
(row-reduce-mat a). Thus, we define

(defund invertiblep (a n) (= (row-rank a) n))

and

(defund inverse-mat (a) (row-reduce-mat a))

First we note that as a consequence of lead-inds-ninit, if (invertiblep a n), then (row-reduce

a) = (id-fmat n):

(defthm row-echelon-p-id-fmat

(implies (and (fmatp a n n) (posp n) (row-echelon-p a) (= (num-nonzero-rows a) n))

(equal a (id-fmat n)))

Now let

p = (inverse-mat a) = (row-reduce-mat a) = (row-ops-mat (row-reduce-ops a) n),

q = (row-ops-mat (invert-row-ops (row-reduce-ops a)) n),

and

r = (fmat* p a) = (row-reduce a).

By invert-row-ops-mat, (fmat* p q) = (fmat* q p) = (id-fmat n). Suppose (row-rank

r) = n. By row-echelon-p-id-fmat, (fmat* p a) = r = (id-fmat n), and by inverse-unique,
a = q. Thus, (invertiblep a n) is a sufficient condition for the existence of an inverse:

(defthmd invertiblep-sufficient

(implies (and (fmatp a n n) (posp n) (invertiblep a n))

(let ((p (inverse-mat a)))

(and (fmatp p n n)

(equal (fmat* a p) (id-fmat n))

(equal (fmat* p a) (id-fmat n))))))

To prove the necessity of (invertiblep a n), suppose (fmatp b n n) and (fmat* a b) = (id-

fmat n). Then

(fmat* r (fmat* b q)) = (fmat* (fmt* p a) (fmat* b q))

= (fmat* p (fmat* (fmat* a b) q))

= (fmat* p q)

= (id-fmat n).

If (invertiblep a n) = NIL, then the last row of r is zero, and the same must be true of (id-fmat
n), a contradiction.

(defthmd invertiblep-necessary

(implies (and (fmatp a n n) (fmatp b n n) (posp n) (= (fmat* a b) (id-fmat n)))

(invertiblep a n)))

We note several consequences of the preceding results. First, an invertible matrix is the inverse of its
inverse:

(defthmd inverse-inverse-mat

(implies (and (fmatp a n n) (posp n) (invertiblep a n))

(and (invertiblep (inverse-mat a) n)

(equal (inverse-mat (inverse-mat a)) a))))

Cancellation laws hold for invertible matrices, e.g.,

(defthmd invertiblep-cancel

(implies (and (fmatp a m n) (fmatp b m n) (fmatp p m m) (posp m) (posp n)

(invertiblep p m))

(iff (equal (fmat* p a) (fmat* p b))

(equal a b))))

A matrix product is invertible iff each factor is invertible:

(defthmd invertiblep-factor

(implies (and (fmatp a n n) (fmatp b n n) (posp n) (invertiblep (fmat* a b) n))

(and (invertiblep a n) (invertiblep b n))))

(defthmd inverse-fmat*

(implies (and (fmatp a n n) (fmatp b n n) (posp n)

(invertiblep a n) (invertiblep b n))

(and (invertiblep (fmat* a b) n)

(equal (inverse-mat (fmat* a b))

(fmat* (inverse-mat b) (inverse-mat a))))))

Finally, we shall show that a is invertible iff its determinant is nonzero. First note that if a has inverse b
and (fdet a) = 0, then by fdet-multiplicative,

(fdet (id-fmat n) n) = (fdet (fmat* a b)) = (f* (fdet a) (fdet b)) = 0,

a contradiction. Thus,

(defthmd invertiblep-fdet-not-zero

(implies (and (fmatp a n n) (posp n) (invertiblep a n))

(not (equal (fdet a n) (f0)))))

On the other hand, assume (fdet a n) ̸= 0. By fmat*-adjoint-fmat,

(fmat* a (adjoint-fmat a n)) = (fmat-scalar-mul (fdet a n) (id-fmat n)),

which implies

(fmat* a (fmat-scalar-mul (f/ (fdet a n)) (adjoint-fmat a n)))

= (fmat-scalar-mul (f/ (fdet a n)) (fmat* a (adjoint-fmat a n)))

= (fmat-scalar-mul (f/ (fdet a n)) (fmat-scalar-mul (fdet a n) (id-fmat n)))

= (id-fmat n),

and by invertiblep-necessary, a is invertible. This also establishes an alternative method for com-
puting the inverse:

(defthmd fdet-not-invertiblep-zero

(implies (and (fmatp a n n) (natp n) (> n 1) (not (equal (fdet a n) (f0))))

(and (invertiblep a n)

(equal (inverse-mat a)

(fmat-scalar-mul (f/ (fdet a n)) (adjoint-fmat a n))))))

3 Simultaneous Systems of Linear equations

Let a be an m×n matrix with (entry i j a) = ai,j for 0 ≤ i< m and 0 ≤ j< n, and let b = (b0 . . .bm−1)
be an flist of length m. We seek an flist x = (x0 . . .xn−1) of length n such that for 0 ≤ i < m,

ai,0x0 + . . .+ai,n−1xn−1 = bi.

We shall refer to a as the coefficient matrix of this system of m linear equations in n unknowns. To express
the system as a matrix equation, we define the column matrix corresponding to a given flist:

(defund col-mat (x) (transpose-mat (list x)))

The above equations are naturally expressed by the matrix equation in the following definition:

(defund solutionp (x a b) (equal (fmat* a (col-mat x)) (col-mat b)))

Let bc = (col-mat b), xc = (col-mat x), p = (row-reduce-mat a), ar = (fmat* p a), and br

= (fmat* p bc). Left-multiplying the above equation by p yields the equivalent equation

(fmat* ar xc) = br. (1)

Thus, we have

(defthmd reduce-linear-equations

(implies (and (fmatp a m n) (posp m) (posp n) (flistnp b m) (flistnp x n))

(let* ((bc (col-mat b)) (xc (col-mat x))

(p (row-reduce-mat a)) (ar (fmat* p a)) (br (fmat* p bc)))

(iff (solutionp x a b)

(equal (fmat* ar xc) br)))))

Our objective, therefore, is to compute an n×1 column matrix xc that solves Equation (1), in which ar

is an m×n reduced row-echelon matrix and br is an m×1 column matrix.
Let q = (num-nonzero-rows ar) = (row-rank a). We shall show that the existence of a solu-

tion to this equation is determined by whether the last m − q entries of br are all 0. This is true iff the
following search returns NIL:

(defun find-nonzero (br q m)

(if (and (natp q) (natp m) (< q m))

(if (= (entry (1- m) 0 br) (f0))

(find-nonzero br q (1- m))

(1- m))

()))

Thus, we define

(defun solvablep (a b)

(null (find-nonzero (fmat* (row-reduce-mat a) (col-mat b))

(row-rank a)

(len a))))

Suppose first that (find-nonzero br q m) = k ̸= NIL, so that (solvablep a b) = NIL. Then (row

k ar) = (flistn0 n) and (entry k 0 br) ̸= 0. It follows that (entry k 0 (fmat* ar xc)) ̸=
(nth k 0 br), and hence (fmat* ar xc) ̸= br. Combining this with reduce-linear-equations,
we conclude that the system of equations has no solution:

(defthmd linear-equations-unsolvable-case

(implies (and (fmatp a m n) (posp m) (posp n) (flistnp b m) (flistnp x n)

(not (solvablep a b)))

(not (solutionp x a b))))

Thus, we may assume (solvablep a b) = T. As a first step toward the solution, consider the
matrices aq and bq consisting of the first q rows of ar and br, respectively, computed by the following:

(defun first-rows (q a)

(if (zp q) ()

(cons (car a) (first-rows (1- q) (cdr a)))))

It is easily shown that aq is a reduced row-echelon q×n matrix of row rank q and that (fmat* ar xc)

= br iff (fmat* aq xc) = bq. Our objective, therefore, is to solve the equation (fmat* aq xc) = bq.

3.1 Uniquely Solvable Case

By row-rank<=n, q ≤ n. We first consider the case q = n. By row-echelon-p-id-fmat, aq =
(id-fmat n) and (fmat* aq xc) = bq iff xc = bq. Combining this observation with first-rows-

linear-equations and reduce-linear-equations, we conclude that there exists a unique solution
in this case:;

(defthmd linear-equations-unique-solution-case

(let* ((br (fmat* (row-reduce-mat a) (col-mat b)))

(bq (first-rows n br)))

(implies (and (fmatp a m n) (posp m) (posp n) (flistnp b m) (flistnp x n)

(solvablep a b) (= (row-rank a) n))

(iff (solutionp x a b)

(equal x (col 0 bq))))))

Our results on cofactor expansion lead to an alternative method of solving a system of n linear
equations in n unknowns in the case of a unique solution, known as Cramer’s rule. Suppose m = n = q,
so that a is an invertible n×n matrix. Our objective is to compute, as a function of a and b, for each i <
n, the ith component (nth i x) of the unique x such that

(fmat* a xc) = bc. (2)

We refer to the analogs of the results of [6, Sec. 5] that appear in fdet.lisp. In particular, we shall sub-
stitute a’ = (replace-row (transpose-mat a) i b) for a in fdot-cofactor-fmat-row-fdet.
Clearly, (row i a’) = b. By cofactor-fmat-transpose,

(cofactor-fmat-row i a’ n) = (cofactor-fmat-row i (transpose-mat a) n)

= (row i (cofactor-fmat (transpose-mat a) n))

= (row i (adjoint-fmat a n)),

and by fdet-transpose,

(fdet a’ n) = (fdet (transpose-fmat (replace-col a i b)) n)

= (fdet (replace-col a i b) n).

Thus, the substitution yields the following:

(fdot b (row i (adjoint-fmat a n))) = (fdet (replace-col a i b) n)).

Multiplying Equation (2) by (adjoint-fmat a n) yields

(fmat* (adjoint-fmat a n) (fmat* a xc)) = (fmat* (adjoint-fmat a n) bc).

But

(fmat* (adjoint-fmat a n) (fmat* a xc))

= (fmat* (fmat* (adjoint-fmat a n) a) xc)

= (fmat* (flist-scalar-mul (fdet a n) (id-fmat n)) xc)

= (flist-scalar-mul (fdet a n) (fmat* (id-fmat n) xc))

= (flist-scalar-mul (fdet a n) xc),

and hence

(flist-scalar-mul (fdet a n) xc) = (fmat* (adjoint-fmat a n) bc).

Equating the entries of these matrices in row i and column 0, we have

(f* (fdet a n) (nth i x)) = (fdot b (row i (adjoint-fmat a n)))

= (fdet (replace-col a i b) n),

which yields Cramer’s rule:

(defthmd cramer

(implies (and (fmatp a n n) (natp n) (> n 1) (invertiblep a n)

(flistnp b n) (flistnp x n) (solutionp x a b)

(natp i) (< i n))

(equal (nth i x)

(f* (f/ (fdet a n))

(fdet (replace-col a i b) n)))))

3.2 General Solvable Case

In the remainder of this section, we treat the general case (solvablep a b) = T with arbitrary q =
(row-rank a) ≤ n. The desired equation (fmat* aq xc) = bq holds iff for 0 ≤ i < q,

(nth i (fmat* aq xc)) = (nth i bq)

or equivalently,

(fdot (row i aq) x) = (car (nth i bq)). (3)

We shall split the dot product (fdot (nth i aq) x) into two sums, corresponding to the list (lead-
inds aq) of leading indices and the list of remaining indices, which we call the free indices:

(defund free-inds (a n) (set-difference-equal (ninit n) (lead-inds a)))

In general, given a sublist inds of (ninit n) and two flists r and x of length n, the following function
extracts and sums the terms of the dot product of r and x that correspond to the indices inds:

(defun fdot-select (inds r x)

(if (consp inds)

(f+ (f* (nth (car inds) r) (nth (car inds) x))

(fdot-select (cdr inds) r x))

(f0)))

In particular, (fdot (row i aq) x) may be expressed as the following sum:

(f+ (fdot-select (lead-inds aq) (row i aq) x)

(fdot-select (free-inds aq n) (row i aq) x))))).

Now since (row i aq) has a 1 at index (nth i (lead-inds aq)) and a 0 at all other lead indices,
the first of these two sums reduces to the single term (nth (nth i (lead-inds aq)) x), and hence
Equation (3) may be expressed as

(nth (nth i (lead-inds aq)) x)

= (f+ (car (nth i bq))

(f- (fdot-select (free-inds aq n) (row i aq) x))).

Thus, x is a solution of our system of equations iff this condition holds for all i < q. This is checked
recursively by the following function:

(defun solution-test-aux (x aq bq lead-inds free-inds k)

(if (zp k) t

(and (equal (nth (nth (1- k) lead-inds) x)

(f+ (car (nth (1- k) bq))

(f- (fdot-select free-inds (nth (1- k) aq) x))))

(solution-test-aux x aq bq lead-inds free-inds (1- k)))))

(defund solution-test (x a b n)

(let* ((ar (row-reduce a))

(br (fmat* (row-reduce-mat a) (col-mat b)))

(q (num-nonzero-rows ar))

(aq (first-rows q ar))

(bq (first-rows q br))

(lead-inds (lead-inds aq))

(free-inds (free-inds aq n)))

(solution-test-aux x aq bq lead-inds free-inds q)))

This provides a test to be applied to a candidate solution:

(defthmd linear-equations-solvable-case

(implies (and (fmatp a m n) (posp m) (posp n) (flistnp b m) (flistnp x n)

(solvablep a b))

(iff (solutionp x a b)

(solution-test x a b n))))

If q = (len (lead-inds aq)) = n, then (free-inds aq n) = NIL, the equation

(nth (nth i l) x) = (f+ (car (nth i bq)) (f- (fdot-select f (nth i aq) x)))

reduces to

(nth i x) = (car (nth i bq),

(solution-test-aux x aq bq l f q) reduces to x = (col 0 bq), and the last theorem reduces
to the earlier result linear-equations-unique-solution-case.

Otherwise, (free-inds aq n) ̸= NIL and the components of x corresponding to the indices in
(lead-inds aq) are determined by the components corresponding to (free-inds aq n), which are
unconstrained. Thus, there is a single solution corresponding to every assignment of values to the latter
set of components, and hence infinitely many solutions. We shall revisit this result in Part III, where
we show that in the homogeneous case, b = (flistn0 n), the solutions form a vector space of dimen-
sion n − q. A basis for this solution space will be provided by a formula derived from the function
solution-test. .

4 Future Work

This formalization of linear algebra is a work in progress. In Part I, we developed the algebra of matrices
over a commutative ring with unity and the theory of determinants. In this sequel, we have restricted our
attention to matrices over a field in order to address the process of row reduction and its applications. To
allow our results to be applied to an arbitrary ring or field, we have characterized each by an encapsulated
set of constrained functions.

There is progress to report on a planned Part III, which begins with another encapsulation that formal-
izes the notion of an abstract finite-dimensional vector space over the field F. The constrained functions
of this encapsulation naturally include a predicate that recognizes vectors in the space, the operations of
vector addition and scalar multiplication, and the constant 0 vector. Two additional functions embody
the requirement of finite dimensionality: (1) a constant list of vectors of unspecified length that serves
as a canonical basis, and (2) a function that returns the coordinates of a given vector with respect to this
basis. Thus, whenever we define a concrete vector space, we are obligated to identify a basis for it. This
establishes a tight connection between vector spaces and matrices: a list of vectors may be identified by
the matrix of coordinates of its members. As an unexpected application of row reduction, this connection
provides an algorithmic definition of the basic notion of linear independence without use of quantifiers:
a list of vectors is linearly independent iff the row rank of its coordinate matrix is the length of the list.

The reader may have noticed that the definition of the basic notion of row equivalence is omitted
from our treatment of row reduction. Recall that two matrices are said to be row equivalent if one may be
derived from the other by a sequence of elementary row operations. This definition could be formalized
in ACL2 using the support for existential quantification provided by defun-sk, and the properties of an
equivalence relation could be derived from the results of Section 2. We could also prove the important
theorem that distinct reduced row-echelon matrices cannot be row equivalent. However, since its most
expedient proof is based on vector spaces (in particular, the row space of a matrix), this result is postponed
to Part III. Note that it provides an alternative definition of row equivalence that avoids quantification:
two matrices a and b are row equivalent iff (row-reduce a) = (row-reduce b). Since we prefer this
algorithmic formulation, the entire topic is deferred to Part III.

Other topics to be addressed in the sequel include linear transformations and diagonalization. As dis-
cussed in Part I, a factor in our decision to develop the algebra of matrices over an arbitrary commutative
ring rather than a field is that this allows us to define the characteristic polynomial of a square matrix
over the field F as the determinant of a certain matrix over the polynomial ring F[t]. A related objective
is the proof of the Cayley-Hamilton Theorem (every square matrix over a commutative ring is a root of
its own characteristic polynomial), which has wide-ranging applications in other areas of mathematics.

This project is part of a broader effort in the formalization of algebra, which began with group
theory [3, 4, 5] and will continue beyond linear algebra. Our next targeted area of investigation will
be Galois theory, which we hope eventually to apply to the study of algebraic number fields. These
intended applications guided our choices of formalization schemes for the basic algebraic structures.
Since we are interested in infinite rings and fields, the encapsulation approach seems to be the only
viable representation scheme for these structures provided by the ACL2 logic. The disadvantages of
not being able to refer to such a structure as an ACL2 object are obvious. On the other hand, since the
groups of primary interest are finite, our investigation of group theory is limited to the finite case. Under
this restriction, a group is conveniently represented as an object characterized by a predicate defined as
an ACL2 function. There are, however, infinite groups of interest, which are not accommodated by our
theory. For example, the general linear group of invertible matrices over a field, which is in general an
infinite structure, would otherwise have provided an interesting example and another connection between
group theory and linear algebra.

Although we have no immediate plans to apply this theory beyond the realm of pure mathematics,
its potential utility in the formal verification of hardware and software applications is limitless. Linear
algebra is central to the rapidly advancing fields of machine learning and neural networks [1], providing
essential tools for data reprersentation and manipulation. Along with finite group theory, it is also im-
portant to a variety of cryptographic algorithms [2]. It is our hope that ACL2 users who are interested in
pursuing such applications may find our results useful.

